

# NAG Library Routine Document

## C05AXF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

C05AXF attempts to locate a zero of a continuous function using a continuation method based on a secant iteration. It uses reverse communication for evaluating the function.

### 2 Specification

```
SUBROUTINE C05AXF (X, FX, TOL, IR, SCAL, C, IND, IFAIL)
  INTEGER          IR, IND, IFAIL
  REAL (KIND=nag_wp) X, FX, TOL, SCAL, C(26)
```

### 3 Description

C05AXF uses a modified version of an algorithm given in Swift and Lindfield (1978) to compute a zero  $\alpha$  of a continuous function  $f(x)$ . The algorithm used is based on a continuation method in which a sequence of problems

$$f(x) - \theta_r f(x_0), \quad r = 0, 1, \dots, m$$

are solved, where  $1 = \theta_0 > \theta_1 > \dots > \theta_m = 0$  (the value of  $m$  is determined as the algorithm proceeds) and where  $x_0$  is your initial estimate for the zero of  $f(x)$ . For each  $\theta_r$  the current problem is solved by a robust secant iteration using the solution from earlier problems to compute an initial estimate.

You must supply an error tolerance TOL. TOL is used directly to control the accuracy of solution of the final problem ( $\theta_m = 0$ ) in the continuation method, and  $\sqrt{\text{TOL}}$  is used to control the accuracy in the intermediate problems ( $\theta_1, \theta_2, \dots, \theta_{m-1}$ ).

### 4 References

Swift A and Lindfield G R (1978) Comparison of a continuation method for the numerical solution of a single nonlinear equation *Comput. J.* **21** 359–362

### 5 Parameters

**Note:** this routine uses **reverse communication**. Its use involves an initial entry, intermediate exits and re-entries, and a final exit, as indicated by the parameter **IND**. Between intermediate exits and re-entries, **all parameters other than FX must remain unchanged**.

1: X – REAL (KIND=nag\_wp) *Input/Output*

*On initial entry:* an initial approximation to the zero.

*On intermediate exit:* the point at which  $f$  must be evaluated before re-entry to the routine.

*On final exit:* the final approximation to the zero.

2: FX – REAL (KIND=nag\_wp) *Input*

*On initial entry:* if  $\text{IND} = 1$ , FX need not be set.

If  $\text{IND} = -1$ , FX must contain  $f(X)$  for the initial value of X.

*On intermediate re-entry:* must contain  $f(X)$  for the current value of X.

3: TOL – REAL (KIND=nag\_wp) Input

*On initial entry:* a value that controls the accuracy to which the zero is determined. TOL is used in determining the convergence of the secant iteration used at each stage of the continuation process. It is used directly when solving the last problem ( $\theta_m = 0$  in Section 3), and  $\sqrt{\text{TOL}}$  is used for the problem defined by  $\theta_r$ ,  $r < m$ . Convergence to the accuracy specified by TOL is not guaranteed, and so you are recommended to find the zero using at least two values for TOL to check the accuracy obtained.

*Constraint:* TOL > 0.0.

4: IR – INTEGER Input

*On initial entry:* indicates the type of error test required, as follows. Solving the problem defined by  $\theta_r$ ,  $1 \leq r \leq m$ , involves computing a sequence of secant iterates  $x_r^0, x_r^1, \dots$ . This sequence will be considered to have converged only if:

for IR = 0,

$$|x_r^{(i+1)} - x_r^{(i)}| \leq eps \times \max(1.0, |x_r^{(i)}|),$$

for IR = 1,

$$|x_r^{(i+1)} - x_r^{(i)}| \leq eps,$$

for IR = 2,

$$|x_r^{(i+1)} - x_r^{(i)}| \leq eps \times |x_r^{(i)}|,$$

for some  $i > 1$ ; here  $eps$  is either TOL or  $\sqrt{\text{TOL}}$  as discussed above. Note that there are other subsidiary conditions (not given here) which must also be satisfied before the secant iteration is considered to have converged.

*Constraint:* IR = 0, 1 or 2.

5: SCAL – REAL (KIND=nag\_wp) Input

*On initial entry:* a factor for use in determining a significant approximation to the derivative of  $f(x)$  at  $x = x_0$ , the initial value. A number of difference approximations to  $f'(x_0)$  are calculated using

$$f'(x_0) \sim (f(x_0 + h) - f(x_0))/h$$

where  $|h| < |\text{SCAL}|$  and  $h$  has the same sign as SCAL. A significance (cancellation) check is made on each difference approximation and the approximation is rejected if insignificant.

*Suggested value:*  $\sqrt{\epsilon}$ , where  $\epsilon$  is the **machine precision** returned by X02AJF.

*Constraint:* SCAL must be sufficiently large that  $X + \text{SCAL} \neq X$  on the computer.

6: C(26) – REAL (KIND=nag\_wp) array Communication Array

(C(5) contains the current  $\theta_r$ , this value may be useful in the event of an error exit.)

7: IND – INTEGER Input/Output

*On initial entry:* must be set to 1 or -1.

IND = 1

FX need not be set.

IND = -1

FX must contain  $f(X)$ .

*On intermediate exit:* contains 2, 3 or 4. The calling program must evaluate  $f$  at  $X$ , storing the result in FX, and re-enter C05AXF with all other parameters unchanged.

*On final exit:* contains 0.

*Constraint:* on entry  $IND = -1, 1, 2, 3$  or 4.

8: IFAIL – INTEGER

*Input/Output*

*On initial entry:* IFAIL must be set to 0,  $-1$  or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value  $-1$  or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, because for this routine the values of the output parameters may be useful even if  $IFAIL \neq 0$  on exit, the recommended value is  $-1$ . **When the value  $-1$  or 1 is used it is essential to test the value of IFAIL on exit.**

*On final exit:*  $IFAIL = 0$  unless the routine detects an error or a warning has been flagged (see Section 6).

## 6 Error Indicators and Warnings

If on entry  $IFAIL = 0$  or  $-1$ , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry,  $TOL \leq 0.0$ ,  
or  $IR \neq 0, 1$  or 2.

IFAIL = 2

The parameter  $IND$  is incorrectly set on initial or intermediate entry.

IFAIL = 3

$SCAL$  is too small, or significant derivatives of  $f$  cannot be computed (this can happen when  $f$  is almost constant and nonzero, for any value of  $SCAL$ ).

IFAIL = 4

The current problem in the continuation sequence cannot be solved, see C(5) for the value of  $\theta_r$ . The most likely explanation is that the current problem has no solution, either because the original problem had no solution or because the continuation path passes through a set of insoluble problems. This latter reason for failure should occur rarely, and not at all if the initial approximation to the zero is sufficiently close. Other possible explanations are that  $TOL$  is too small and hence the accuracy requirement is too stringent, or that  $TOL$  is too large and the initial approximation too poor, leading to successively worse intermediate solutions.

IFAIL = 5

Continuation away from the initial point is not possible. This error exit will usually occur if the problem has not been properly posed or the error requirement is extremely stringent.

IFAIL = 6

The final problem (with  $\theta_m = 0$ ) cannot be solved. It is likely that too much accuracy has been requested, or that the zero is at  $\alpha = 0$  and  $IR = 2$ .

IFAIL =  $-99$

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.8 in the Essential Introduction for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.  
See Section 3.7 in the Essential Introduction for further information.

IFAIL = -999

Dynamic memory allocation failed.  
See Section 3.6 in the Essential Introduction for further information.

## 7 Accuracy

The accuracy of the approximation to the zero depends on TOL and IR. In general decreasing TOL will give more accurate results. Care must be exercised when using the relative error criterion (IR = 2).

If the zero is at  $X = 0$ , or if the initial value of  $X$  and the zero bracket the point  $X = 0$ , it is likely that an error exit with IFAIL = 4, 5 or 6 will occur.

It is possible to request too much or too little accuracy. Since it is not possible to achieve more than machine accuracy, a value of TOL  $\ll$  *machine precision* should not be input and may lead to an error exit with IFAIL = 4, 5 or 6. For the reasons discussed under IFAIL = 4 in Section 6, TOL should not be taken too large, say no larger than TOL = 1.0E-3.

## 8 Parallelism and Performance

Not applicable.

## 9 Further Comments

For most problems, the time taken on each call to C05AXF will be negligible compared with the time spent evaluating  $f(x)$  between calls to C05AXF. However, the initial value of  $X$  and the choice of TOL will clearly affect the timing. The closer that  $X$  is to the root, the less evaluations of  $f$  required. The effect of the choice of TOL will not be large, in general, unless TOL is very small, in which case the timing will increase.

## 10 Example

This example calculates a zero of  $x - e^{-x}$  with initial approximation  $x_0 = 1.0$ , and TOL = 1.0E-3 and 1.0E-4.

### 10.1 Program Text

```

Program c05axfe

!      C05AXF Example Program Text

!      Mark 25 Release. NAG Copyright 2014.

!      .. Use Statements ..
      Use nag_library, Only: c05axf, nag_wp, x02ajf
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nout = 6
!      .. Local Scalars ..
      Real (Kind=nag_wp)          :: fx, scal, tol, x
      Integer                     :: i, ifail, ind, ir
!      .. Local Arrays ..
      Real (Kind=nag_wp)          :: c(26)
!      .. Intrinsic Procedures ..
      Intrinsic                   :: exp, sqrt
!      .. Executable Statements ..

```

```

Write (nout,*) 'C05AXF Example Program Results'

scal = sqrt(x02ajf())
ir = 0

loop: Do i = 3, 4
  tol = 10.0E0_nag_wp**(-i)
  Write (nout,*)
  Write (nout,99999) 'TOL =', tol
  Write (nout,*)
  x = 1.0E0_nag_wp
  ind = 1
  ifail = -1

revcomm: Do
  Call c05axf(x,fx,tol,ir,scal,c,ind,ifail)

  If (ind==0) Then
    Exit revcomm
  End If

  fx = x - exp(-x)
End Do revcomm

Select Case (ifail)
Case (:-1)
  Exit loop
Case (4,6)
  Write (nout,99998) 'Final value = ', x, ' THETA = ', c(5)
Case (0)
  Write (nout,99998) 'Root is ', x
End Select

End Do loop

99999 Format (1X,A,E11.4)
99998 Format (1X,A,F14.5,A,F10.2)
End Program c05axfe

```

## 10.2 Program Data

None.

## 10.3 Program Results

```

C05AXF Example Program Results

TOL = 0.1000E-02

Root is      0.56715

TOL = 0.1000E-03

Root is      0.56715

```

---